;; -*- coding: utf-8 -*-
(eval-when (:compile-toplevel :load-toplevel :execute)
  (defparameter OPT
    #+swank '(optimize (speed 3) (safety 2))
    #-swank '(optimize (speed 3) (safety 0) (debug 0)))
  #+swank (progn (ql:quickload '(:cl-debug-print :fiveam))
                 (shadow :run)
                 (use-package :fiveam))
  #-swank (set-dispatch-macro-character #\# #\> (lambda (s c p) (declare (ignore c p)) (read s nil (values) t))))
#+swank (cl-syntax:use-syntax cl-debug-print:debug-print-syntax)

;; BEGIN_INSERTED_CONTENTS
(declaim (ftype (function * (values fixnum &optional)) read-fixnum))
(defun read-fixnum (&optional (in *standard-input*))
  (declare #.OPT)
  (macrolet ((%read-byte ()
               `(the (unsigned-byte 8)
                     #+swank (char-code (read-char in nil #\Nul))
                     #-swank (sb-impl::ansi-stream-read-byte in nil #.(char-code #\Nul) nil))))
    (let* ((minus nil)
           (result (loop (let ((byte (%read-byte)))
                           (cond ((<= 48 byte 57)
                                  (return (- byte 48)))
                                 ((zerop byte) ; #\Nul
                                  (error "Read EOF or #\Nul."))
                                 ((= byte #.(char-code #\-))
                                  (setf minus t)))))))
      (declare ((integer 0 #.most-positive-fixnum) result))
      (loop
        (let* ((byte (%read-byte)))
          (if (<= 48 byte 57)
              (setq result (+ (- byte 48) (* 10 (the (integer 0 #.(floor most-positive-fixnum 10)) result))))
              (return (if minus (- result) result))))))))

(defmacro dbg (&rest forms)
  #+swank
  (if (= (length forms) 1)
      `(format *error-output* "~A => ~A~%" ',(car forms) ,(car forms))
      `(format *error-output* "~A => ~A~%" ',forms `(,,@forms)))
  #-swank (declare (ignore forms)))

(defmacro define-int-types (&rest bits)
  `(progn
     ,@(mapcar (lambda (b) `(deftype ,(intern (format nil "UINT~A" b)) () '(unsigned-byte ,b))) bits)
     ,@(mapcar (lambda (b) `(deftype ,(intern (format nil "INT~A" b)) () '(signed-byte ,b))) bits)))
(define-int-types 2 4 7 8 15 16 31 32 62 63 64)

(declaim (inline println))
(defun println (obj &optional (stream *standard-output*))
  (let ((*read-default-float-format* 'double-float))
    (prog1 (princ obj stream) (terpri stream))))

(defconstant +mod+ 1000000007)

;; Body
;;;
;;; 2-dimensional binary indexed tree
;;;

;; TODO: binary search
(defmacro define-2d-bitree (name &key (operator '#'+) (identity 0) sum-type)
  "OPERATOR := binary operator (on a commutative monoid)
IDENTITY := object (identity element of the monoid)
SUM-TYPE := nil | type specifier

Defines no structure; 2D BIT is just a 2-dimensional array. This macro defines
the three functions: <NAME>-UPDATE!, <NAME>-SUM and COERCE-TO-<NAME>!.

SUM-TYPE is used only for the type declaration: each sum is declared to be this
type. (The element-type of vector itself doesn't need to be SUM-TYPE.)"
  (let* ((name (string name))
         (fname-update (intern (format nil "~A-UPDATE!" name)))
         (fname-sum (intern (format nil "~A-SUM" name)))
         (fname-coerce (intern (format nil "COERCE-TO-~A!" name))))
    `(progn
       (declaim (inline ,fname-update))
       (defun ,fname-update (bitree index1 index2 delta)
         "Destructively increments the array: array[INDEX1][INDEX2] += DELTA."
         (let ((length1 (array-dimension bitree 0))
               (length2 (array-dimension bitree 1)))
           (do ((i index1 (logior i (+ i 1))))
               ((>= i length1))
             (declare ((integer 0 #.most-positive-fixnum) i))
             (do ((j index2 (logior j (+ j 1))))
                 ((>= j length2))
               (setf (aref bitree i j)
                     (funcall ,operator (aref bitree i j) delta))))
           bitree))
       
       (declaim (inline ,fname-sum))
       (defun ,fname-sum (bitree end1 end2)
         "Returns the sum of the rectangle region: array[0][0] + ... +
array[END1-1][END2-1]."
         (declare ((integer 0 #.most-positive-fixnum) end1 end2))
         (let ((res ,identity))
           ,@(when sum-type `((declare (type ,sum-type res))))
           (do ((i (- end1 1) (- (logand i (+ i 1)) 1)))
               ((< i 0))
             (declare ((integer -1 #.most-positive-fixnum) i))
             (do ((j (- end2 1) (- (logand j (+ j 1)) 1)))
                 ((< j 0))
               (setf res (funcall ,operator res (aref bitree i j)))))
           res))

       (declaim (inline ,fname-coerce))
       (defun ,fname-coerce (array)
         "Destructively constructs 2D BIT from 2D array."
         (declare (ignorable array))
         (error "Not implemented yet.")))))

(define-2d-bitree bitree
  :operator (lambda (x y) (mod (+ x y) +mod+))
  :identity 0
  :sum-type uint62)

(defun main ()
  (declare #.OPT)
  (let* ((n (read))
         (m (read))
         (ss (make-array n :element-type 'uint31))
         (ts (make-array m :element-type 'uint31))
         (bitree (make-array (list n m) :element-type 'uint31 :initial-element 0)))
    (declare (uint16 n m))
    (dotimes (i n) (setf (aref ss i) (read-fixnum)))
    (dotimes (i m) (setf (aref ts i) (read-fixnum)))
    (dotimes (i n)
      (dotimes (j m)
        (when (= (aref ss i) (aref ts j))
          (bitree-update! bitree i j (+ 1 (bitree-sum bitree i j))))))
    (println (mod (+ 1 (bitree-sum bitree n m)) +mod+))))

#-swank(main)


Measured memory_bytes: 35304